**ABSTRACT**

In the pharmaceutical industry, there are often rules for displaying the number of decimal places in a report. This is particularly important when reporting laboratory results in a table. For example, for a particular laboratory test, if the laboratory result in the data is carried to the tenths place, the displayed mean will be reported at the hundredths place. The common rule is displaying the mean result one place higher than what is in the data. There are similar rules for other statistics such as the median, standard deviation, minimum, and maximum. This paper will illustrate a simple technique using the PUTN function to dynamically set the decimal precision. The techniques presented offer a good overview of basic SAS® functions that will educate programmers at all levels.

**INTRODUCTION**

Summarizing statistics on a continuous variable is a common task in clinical trials computing. Below is a popular way to display these statistics:

<table>
<thead>
<tr>
<th>Parameter (unit)</th>
<th>Treatment A</th>
<th>Treatment B</th>
<th>Placebo</th>
</tr>
</thead>
<tbody>
<tr>
<td>Statistic</td>
<td>N = XXX</td>
<td>N = XXX</td>
<td>N = XXX</td>
</tr>
<tr>
<td>Temperature (°F)</td>
<td></td>
<td></td>
<td></td>
</tr>
<tr>
<td>n</td>
<td>XXX</td>
<td>XXX</td>
<td>XXX</td>
</tr>
<tr>
<td>Mean</td>
<td>XXX.XX</td>
<td>XXX.XX</td>
<td>XXX.XX</td>
</tr>
<tr>
<td>SD</td>
<td>XXX.XX</td>
<td>XXX.XX</td>
<td>XXX.XX</td>
</tr>
<tr>
<td>Median</td>
<td>XXX.XX</td>
<td>XXX.XX</td>
<td>XXX.XX</td>
</tr>
<tr>
<td>Range (Min, Max)</td>
<td>(XXX, XXX)</td>
<td>(XXX, XXX)</td>
<td>(XXX, XXX)</td>
</tr>
</tbody>
</table>

Table 1. Vital signs shell

At a glance, the number of decimal places appears to be fixed across the different statistics within the shell. Programmers need to be attentive to the decimals for each parameter. Temperature is usually presented as XXX.X (98.6°F for example rather than XX F). Therefore it makes sense to display these statistics according to a set of rules. The remainder of this paper will only concentrate on the statistics: mean, standard deviation, median and min/max – clearly sample size (n) is always presented as a whole number.

**RULES FOR SETTING PRECISION**

The common rule is this:

1. Mean and median: the number of decimals is displayed as one more than what is shown in the data.
2. Standard deviation is displayed as two more than what is shown in the data.
3. Min/max are displayed exactly as what is shown in the data (zero places).

For example:

<table>
<thead>
<tr>
<th>TEMP</th>
<th>TRT01PN</th>
</tr>
</thead>
<tbody>
<tr>
<td>97.8</td>
<td>A</td>
</tr>
<tr>
<td>101.3</td>
<td>A</td>
</tr>
<tr>
<td>98.5</td>
<td>B</td>
</tr>
<tr>
<td>99.4</td>
<td>B</td>
</tr>
<tr>
<td>97.2</td>
<td>Placebo</td>
</tr>
</tbody>
</table>

Table 2. Dummy temperature data
The statistics:

<table>
<thead>
<tr>
<th>Parameter (unit)</th>
<th>Treatment A</th>
<th>Treatment B</th>
<th>Placebo</th>
</tr>
</thead>
<tbody>
<tr>
<td>n</td>
<td>2</td>
<td>2</td>
<td>1</td>
</tr>
<tr>
<td>Mean</td>
<td>99.55</td>
<td>98.95</td>
<td>97.20</td>
</tr>
<tr>
<td>SD</td>
<td>2.475</td>
<td>0.636</td>
<td>.</td>
</tr>
<tr>
<td>Median</td>
<td>99.55</td>
<td>98.95</td>
<td>97.20</td>
</tr>
<tr>
<td>Range (Min, Max)</td>
<td>(97.8, 101.3)</td>
<td>(98.5, 99.4)</td>
<td>(97.2, 97.2)</td>
</tr>
</tbody>
</table>

**Table 3: Statistics on dummy temperature data**

The dummy data on temperature goes out to one decimal place. So the mean and median are +1 decimal places equaling two decimal places (99.55 for example), standard deviation is +2 equaling three decimal places (2.475 for example) and min/max are +0 equaling one decimal place (97.8 and 101.3). Note for assigning precision, we can formulate a method with three rules – +0, +1 and +2 scenarios.

**SAS CODE**

The first step is to identify the number of decimal places shown in the data and obtain the statistics needed for reporting. Writing a program to return the precision from the data is ideal, but that is beyond the scope of this paper.

```sas
DATA ex1;
input paramcd $ aval trt01pn $;
cards;
  TEMP 97.8   A
  TEMP 101.3 A
  TEMP 98.5   B
  TEMP 99.4   B
  TEMP 97.2 Placebo
  PULSE 83   B
  PULSE 85 Placebo
  PULSE 88   A
  PULSE 89   B
  PULSE 80 Placebo
;RUN;

PROC MEANS noprint nway;
  class paramcd trt01pn;
  var aval;
  output out = stats
    n   = n
    mean = mean
    std  = std
    median = median
    min  = min
    max  = max;
RUN;
```

Here are the raw values as output from PROC MEANS – assume we have merged on the precision values calculated a priori:

<table>
<thead>
<tr>
<th>paramcd</th>
<th>trt01pn</th>
<th>n</th>
<th>mean</th>
<th>std</th>
<th>median</th>
<th>min</th>
<th>max</th>
<th>prec</th>
</tr>
</thead>
<tbody>
<tr>
<td>PULSE</td>
<td>A</td>
<td>1</td>
<td>88.00</td>
<td>.</td>
<td>88.0</td>
<td>88.0</td>
<td>88.0</td>
<td>0</td>
</tr>
<tr>
<td>PULSE</td>
<td>B</td>
<td>2</td>
<td>86.00</td>
<td>4.24264</td>
<td>86.0</td>
<td>83.0</td>
<td>89.0</td>
<td>0</td>
</tr>
<tr>
<td>PULSE</td>
<td>Placebo</td>
<td>2</td>
<td>82.50</td>
<td>3.53553</td>
<td>82.5</td>
<td>80.0</td>
<td>85.0</td>
<td>0</td>
</tr>
<tr>
<td>TEMP</td>
<td>A</td>
<td>2</td>
<td>99.55</td>
<td>2.47487</td>
<td>99.55</td>
<td>97.8</td>
<td>101.3</td>
<td>1</td>
</tr>
</tbody>
</table>
Dynamically Setting Decimal Precision Using PUTN, continued

<table>
<thead>
<tr>
<th>paramcd</th>
<th>trt01pn</th>
<th>n</th>
<th>mean</th>
<th>std</th>
<th>median</th>
<th>min</th>
<th>max</th>
<th>prec</th>
</tr>
</thead>
<tbody>
<tr>
<td>TEMP</td>
<td>B</td>
<td>2</td>
<td>98.95</td>
<td>0.63640</td>
<td>98.95</td>
<td>98.5</td>
<td>99.4</td>
<td>1</td>
</tr>
<tr>
<td>TEMP</td>
<td>Placebo</td>
<td>1</td>
<td>97.20</td>
<td>.</td>
<td>97.20</td>
<td>97.2</td>
<td>97.2</td>
<td>1</td>
</tr>
</tbody>
</table>

Table 4: Output on dummy vital signs data

The obvious solution is the use of IF/THEN statements. This is a perfectly valid technique and works as follows:

```sas
DATA prec1;
  set stats;
  n_    = put(n,3.);
  if prec = 0 then do;
    mean_ = put(mean,8.1); /*+1*/
    if n > 0 then stddv_ = put(std,8.2); /*+2*/
    med_  = put(median,8.1); /*+1*/
    mnmx_ = '('||strip(put(min,8.))||', '||strip(put(max,8.))||')'; /*+0*/
  end;
  else if prec = 1 then do;
    mean_ = put(mean,8.2);
    if n > 0 then stddv_ = put(std,8.3);
    med_  = put(median,8.2);
    mnmx_ = '('||strip(put(min,8.1))||', '||strip(put(max,8.1))||')';
  end;
RUN;
```

Note these data have a maximum precision of one. If the precision exceeded one, additional IF/THEN statements are needed. In clinical trials, we often analyze laboratory values. These values usually have a wider variance of precision values. Using PUTN, you can dynamically account for any precision value.

**USING PUTN**

The first step is to create three variables for each +1, +2 and +0 scenario.

```sas
DATA prec1;
  set stats;
  p0 = put(sum(prec,0),8.);
p1 = put(sum(prec,1),8.);
p2 = put(sum(prec,2),8.);
RUN;
```

```
Table 5: Output on dummy vital signs data with initial variables
```

Now we have the ingredients to use the PUTN function. The PUTN function allows programmers to pass a variable in place of a format. The next step is to create a variable that has a format value. – e.g., `var = 8.1`. Adding a little more code to our data step:
DATA prec1;
   length prec0-prec2 $8;
   set stats;

   p0 = put(sum(prec,0),8.);
   p1 = put(sum(prec,1),8.);
   p2 = put(sum(prec,2),8.);

   prec0 = cats("8.",p0);
   prec1 = cats("8.",p1);
   prec2 = cats("8.",p2);

RUN;

Now we have:

<table>
<thead>
<tr>
<th>paramcd</th>
<th>trt01pn</th>
<th>n</th>
<th>mean</th>
<th>std</th>
<th>median</th>
<th>min</th>
<th>max</th>
<th>prec</th>
<th>p0</th>
<th>p1</th>
<th>p2</th>
<th>prec0</th>
<th>prec1</th>
<th>prec2</th>
</tr>
</thead>
<tbody>
<tr>
<td>PULSE</td>
<td>A</td>
<td>1</td>
<td>88.00</td>
<td>.</td>
<td>88.00</td>
<td>88.0</td>
<td>0</td>
<td>0</td>
<td>1</td>
<td>2</td>
<td>8.0</td>
<td>8.1</td>
<td>8.2</td>
<td></td>
</tr>
<tr>
<td>PULSE</td>
<td>B</td>
<td>2</td>
<td>86.00</td>
<td>4.24264</td>
<td>86.00</td>
<td>83.0</td>
<td>89.0</td>
<td>0</td>
<td>0</td>
<td>1</td>
<td>2</td>
<td>8.0</td>
<td>8.1</td>
<td>8.2</td>
</tr>
<tr>
<td>PULSE</td>
<td>Placebo</td>
<td>2</td>
<td>82.50</td>
<td>3.53553</td>
<td>82.50</td>
<td>80.0</td>
<td>85.0</td>
<td>0</td>
<td>0</td>
<td>1</td>
<td>2</td>
<td>8.0</td>
<td>8.1</td>
<td>8.2</td>
</tr>
<tr>
<td>TEMP</td>
<td>A</td>
<td>2</td>
<td>99.55</td>
<td>2.47487</td>
<td>99.55</td>
<td>97.8</td>
<td>101.3</td>
<td>1</td>
<td>1</td>
<td>2</td>
<td>3</td>
<td>8.1</td>
<td>8.2</td>
<td>8.3</td>
</tr>
<tr>
<td>TEMP</td>
<td>B</td>
<td>2</td>
<td>98.95</td>
<td>0.63640</td>
<td>98.95</td>
<td>98.5</td>
<td>99.4</td>
<td>1</td>
<td>1</td>
<td>2</td>
<td>3</td>
<td>8.1</td>
<td>8.2</td>
<td>8.3</td>
</tr>
<tr>
<td>TEMP</td>
<td>Placebo</td>
<td>1</td>
<td>97.20</td>
<td>.</td>
<td>97.20</td>
<td>97.2</td>
<td>97.2</td>
<td>1</td>
<td>1</td>
<td>2</td>
<td>3</td>
<td>8.1</td>
<td>8.2</td>
<td>8.3</td>
</tr>
</tbody>
</table>

Table 6: Output on dummy vital signs data with precision variables

Finally we are in position to use the PUTN function.

DATA prec1;
   length prec0-prec2 $8;
   set stats;

   p0 = put(sum(prec,0),8.);
   p1 = put(sum(prec,1),8.);
   p2 = put(sum(prec,2),8.);

   prec0 = cats("8.",p0);
   prec1 = cats("8.",p1);
   prec2 = cats("8.",p2);

   n_    = put(n,3.);
   mean_ = putn(mean,prec1);  /*+1*/
   if n > 0 then stddv_ = putn(std,prec2);  /*+2*/
   med_  = putn(median,prec1);  /*+1*/
   mnmx_ = '('||strip(putn(min,prec0))||', '||strip(putn(max,prec0))||')';  /*+0*/

RUN;

We can also create the variables in arrays – an even more dynamic approach:

DATA prec1;
   length prc0-prc2 $8;
   set stats;

   array p{3} $ p0-p2;
   array prc{3} prc0-prc2;
   do i = 1 to 3;
p(i) = put(sum(prec,i - 1),8.);
prc(i) = cats("8.",p(i));

n_ = put(n,3.);
mean_ = putn(mean,prc1);
if n > 0 then stddv_ = putn(std, prc2);
med_ = putn(median, prc1);
mnmx_ = '('||strip(putn(min,prc0))||', '||strip(putn(max,prc0))||')';
end;

RUN;

<table>
<thead>
<tr>
<th>paramcd</th>
<th>trt01pn</th>
<th>prec</th>
<th>n_</th>
<th>mean_</th>
<th>stddv_</th>
<th>med_</th>
<th>mnmx_</th>
</tr>
</thead>
<tbody>
<tr>
<td>PULSE</td>
<td>A</td>
<td>0</td>
<td>1</td>
<td>88.0</td>
<td>.</td>
<td>88.0</td>
<td>(88, 88)</td>
</tr>
<tr>
<td>PULSE</td>
<td>B</td>
<td>0</td>
<td>2</td>
<td>86.0</td>
<td>4.24</td>
<td>86.0</td>
<td>(83, 89)</td>
</tr>
<tr>
<td>PULSE</td>
<td>Placebo</td>
<td>0</td>
<td>2</td>
<td>82.5</td>
<td>3.54</td>
<td>82.5</td>
<td>(80, 85)</td>
</tr>
<tr>
<td>TEMP</td>
<td>A</td>
<td>1</td>
<td>2</td>
<td>99.55</td>
<td>2.475</td>
<td>99.55</td>
<td>(97.8, 101.3)</td>
</tr>
<tr>
<td>TEMP</td>
<td>B</td>
<td>1</td>
<td>2</td>
<td>98.95</td>
<td>0.636</td>
<td>98.95</td>
<td>(98.5, 99.4)</td>
</tr>
<tr>
<td>TEMP</td>
<td>Placebo</td>
<td>1</td>
<td>1</td>
<td>97.20</td>
<td>.</td>
<td>97.20</td>
<td>(97.2, 97.2)</td>
</tr>
</tbody>
</table>

Table 7: Final output on dummy vital signs data with precision variables

A simple transpose creates a summary table:

<table>
<thead>
<tr>
<th>paramcd</th>
<th><em>NAME</em></th>
<th>A</th>
<th>B</th>
<th>Placebo</th>
</tr>
</thead>
<tbody>
<tr>
<td>PULSE</td>
<td>n_</td>
<td>1</td>
<td>2</td>
<td>2</td>
</tr>
<tr>
<td>PULSE</td>
<td>mean_</td>
<td>88.0</td>
<td>86.0</td>
<td>82.5</td>
</tr>
<tr>
<td>PULSE</td>
<td>stddv_</td>
<td>.</td>
<td>4.24</td>
<td>3.54</td>
</tr>
<tr>
<td>PULSE</td>
<td>med_</td>
<td>88.0</td>
<td>86.0</td>
<td>82.5</td>
</tr>
<tr>
<td>PULSE</td>
<td>mnmx_</td>
<td>(88, 88)</td>
<td>(83, 89)</td>
<td>(80, 85)</td>
</tr>
<tr>
<td>TEMP</td>
<td>n_</td>
<td>2</td>
<td>2</td>
<td>1</td>
</tr>
<tr>
<td>TEMP</td>
<td>mean_</td>
<td>99.55</td>
<td>98.95</td>
<td>97.20</td>
</tr>
<tr>
<td>TEMP</td>
<td>stddv_</td>
<td>2.475</td>
<td>0.636</td>
<td>.</td>
</tr>
<tr>
<td>TEMP</td>
<td>med_</td>
<td>99.55</td>
<td>98.95</td>
<td>97.20</td>
</tr>
<tr>
<td>TEMP</td>
<td>mnmx_</td>
<td>(97.8, 101.3)</td>
<td>(98.5, 99.4)</td>
<td>(97.2, 97.2)</td>
</tr>
</tbody>
</table>

Table 8: Final transposed output on dummy vital signs data with precision variables

CONCLUSION

The PUTN function offers a simple solution to assigning precision. Without using this function, programmers are left with other techniques (often IF/THEN statements), which arrive at the same results. As the precision values vary (as we see in laboratory data), more IF/THEN statements are required. However adding more code clutters the program and copying/pasting more IF/THEN statements is prone to error. Using PUTN reduces the amount of code and relies on the data to dictate the number of decimal places.

ACKNOWLEDGMENTS

Eva J. Welch
John Ingersoll
Steve Noga
CONTACT INFORMATION

Your comments and questions are valued and encouraged. Contact the author at:

Brandon Welch
Rho, Inc
6330 Quadrangle Dr., Ste. 500
Chapel Hill, NC 27517
Phone: 919-595-6592
Fax: 919-408-0999
Email: Brandon_Welch@rhoworld.com
Web: www.rhoworld.com

SAS and all other SAS Institute Inc. product or service names are registered trademarks or trademarks of SAS Institute Inc. in the USA and other countries. ® indicates USA registration.

Other brand and product names are registered trademarks or trademarks of their respective companies.